
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.2.2.9, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_2_2_9
Test Group:
RLC

ATS Version:
iwd-EUTRA-TVB2009-03_D09wk51
System Simulator used:
Anite CT
UE used:
LG Electronics LEO3 UE.

Verification Status:
PASS
4 Corrections required for test case 7.2.2.9

4.1 Introduction

This section describes the changes required to make test case 7.2.2.9 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D09wk51 release.

4.2 Change 1

	Testcase name
	f_TC_7_2_2_9_EUTRA

	Reason for change
	As per the Prose the one AM DRB and one UM DRB has to be setup. 

	Summary of change
	Changed to add on default AM DRB

	Source of change
	File : RLC_UM_Testcases.ttcn


Before:

	function f_TC_7_2_2_9_EUTRA()

runs on EUTRA_PTC {

  var RLC_SS_State v_RLC_Rec;

  var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

  var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

  // Initialization, call setup, DRB loopback

  f_EUTRA_Init(c1);

  f_InitRLC_Record(v_RLC_Rec, UM_Mode);

  // Cell configuration, with one UM DRB at the SS side

  f_EUTRA_CellConfig_DefPlusUM (eutra_Cell1);

  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

  f_EUTRA_GenericRbEst_WithRLC_UM_Params (eutra_Cell1,

                                          {cs_DRB_ToAddMod_RLC_UM (tsc_DRB2,

                                                                   cs_508_PDCP_Config_DRB_UM,

                                                                   cs_508_DRB_UL_UM_RLC,

                                                                   cs_RX_UM_RLC_DRB_SN10_tReord200ms)});
  // reconfigure DRB locally to set the required test mode UM mapped to TM

  f_EUTRA_SS_DRB2_ConfigUM_TM_Mode ( eutra_Cell1);

  //closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);

  f_EUTRA_TestBody_Set(true);

  fl_7_2_2_9_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  //postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, ATTACHED );

} // f_TC_7_2_2_9_EUTRA


After: 

	function f_TC_7_2_2_9_EUTRA()

runs on EUTRA_PTC {

  var RLC_SS_State v_RLC_Rec;

  var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

  var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

  var template (omit) DRB_ToAddModList v_DRB_ToAddModList:= omit;
  // Initialization, call setup, DRB loopback

  f_EUTRA_Init(c1);

  f_InitRLC_Record(v_RLC_Rec, UM_Mode);

  // Cell configuration, with one UM DRB at the SS side

  f_EUTRA_CellConfig_DefPlusUM (eutra_Cell1);

  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1);

  v_DRB_ToAddModList[1] :=  cs_DRB_ToAddMod_RLC_UM (tsc_DRB2,

                                                    cs_508_PDCP_Config_DRB_UM,

                                                    cs_508_DRB_UL_UM_RLC,

                                                    cs_RX_UM_RLC_DRB_SN10_tReord200ms);

  //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

  f_EUTRA_GenericRbEst_WithRLC_UM_Params (eutra_Cell1,

                                          v_DRB_ToAddModList);
  // reconfigure DRB locally to set the required test mode UM mapped to TM

  f_EUTRA_SS_DRB2_ConfigUM_TM_Mode ( eutra_Cell1);

  //closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);

  f_EUTRA_TestBody_Set(true);

  fl_7_2_2_9_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  //postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, ATTACHED );

} // f_TC_7_2_2_9_EUTRA


4.3 Change 2

	Testcase name
	fl_7_2_2_9_TestBody

	Reason for change
	1) UMD PDU #8 contains the last part of SDU #4. Hence the FI should indicate the last part of SDU.

2) As per 36.508 timers should have 10% tolerance. In the TTCN at the moment tolerance is not implemented.

3) Step 10 and step 11 will be the first and second PDU (SDU #2 and SDU #4) being transmitted from UE’s PDCP entity for DRB2. Hence the PDCP sequence number for these PDU will be 0 and 1 respectively. 

4) SDU#2 and SDU #4, when looped back from UE will have the PDCP SN as 0 and 1. Hence the SDU #2 and SDU #4 should be constructed with data of that of SDU#2 and SDU #4 but with PDCP SN 0 and 1. 

	Summary of change
	1) Changed the FI to indicate that its last part of SDU.

2) Added the tolerance of 10% for the t-reordering timers.

3) Changed the PDU, with data as that of SDU #2 and SDU #4 and PDCP SN of 0 and 1 respectively.

4) Added the SDU#5 and SDU#6, which has PDCP data as that of SDU #2 and SDU #4 but the PDCP SN 0 and 1 respectively.

	Source of change
	File : RLC_UM_Testcases.ttcn


Before:

	function fl_7_2_2_9_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC

{

  timer t_TReordering := 0.2; // seconds

  var integer v_RLCSDUsize := 128;

  fl_GenerateRLC_UM_SDUs (p_RLC_Rec, v_RLCSDUsize-1, 4); // PDCP SDU size = RLCSDU size - 1 octet (PDCP header)
  // Step 1. The SS transmits UMD PDU#1 containing first segment of RLC SDU#1.

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU1, 0, 75, tsc_FI_StartOfSDU);

  // Step 2. The SS does NOT transmit UMD PDU#2 containing last segment of RLC SDU#1.

  // Step 3. The SS transmits UMD PDU#3 containing first segment of RLC SDU#2.

  p_RLC_Rec.UM_VTUS  := 2;

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU2, 0, 75, tsc_FI_StartOfSDU);

  // Step 4. The SS transmits UMD PDU#4 containing last segment of RLC SDU#2.

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU2, 75, v_RLCSDUsize - 75, tsc_FI_EndOfSDU);

  // Step 5. The SS transmits UMD PDU#5 containing first segment of RLC SDU#3.

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU3, 0, 75, tsc_FI_StartOfSDU);

  // Step 6. The SS does NOT transmit UMD PDU#6 containing last segment of RLC SDU#3.

  // (Just get the data to update data pointers in PRBS)

  // Step 7. The SS transmits UMD PDU#7 containing first segment of RLC SDU#4.

  p_RLC_Rec.UM_VTUS  := 6;

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU4, 0, 75, tsc_FI_StartOfSDU);

  // Step 8. The SS transmits UMD PDU#8 containing last segment of RLC SDU#4.

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU4, 75, v_RLCSDUsize - 75, tsc_FI_StartOfSDU);

  // Step 9. Wait for 200 ms to ensure that t_Reordering for the UMD PDU#2 expires.

  f_Delay (0.2);

  // Step 10. Check: Does the UE loop back the RLC SDU#2.

  f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU2 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  // Step 11. Wait for 200 ms to ensure that t_Reordering for the UMD PDU#6 expires.

  f_Delay (0.2);
  // Step 12. Check: Does the UE loop back the RLC SDU#4.

  f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU4 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

} // fl_7_2_2_9_TestBody




After: 

	function fl_7_2_2_9_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC

{

  timer t_TReordering := 0.2; // seconds

  var integer v_RLCSDUsize := 128;

  fl_GenerateRLC_UM_SDUs_7229 (p_RLC_Rec, v_RLCSDUsize-1, 4); // PDCP SDU size = RLCSDU size - 1 octet (PDCP header) //NOTE 4
  // Step 1. The SS transmits UMD PDU#1 containing first segment of RLC SDU#1.

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU1, 0, 75, tsc_FI_StartOfSDU);

  // Step 2. The SS does NOT transmit UMD PDU#2 containing last segment of RLC SDU#1.

  // Step 3. The SS transmits UMD PDU#3 containing first segment of RLC SDU#2.

  p_RLC_Rec.UM_VTUS  := 2;

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU2, 0, 75, tsc_FI_StartOfSDU);

  // Step 4. The SS transmits UMD PDU#4 containing last segment of RLC SDU#2.

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU2, 75, v_RLCSDUsize - 75, tsc_FI_EndOfSDU);

  // Step 5. The SS transmits UMD PDU#5 containing first segment of RLC SDU#3.

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU3, 0, 75, tsc_FI_StartOfSDU);

  // Step 6. The SS does NOT transmit UMD PDU#6 containing last segment of RLC SDU#3.

  // (Just get the data to update data pointers in PRBS)

  // Step 7. The SS transmits UMD PDU#7 containing first segment of RLC SDU#4.

  p_RLC_Rec.UM_VTUS  := 6;

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU4, 0, 75, tsc_FI_StartOfSDU);

  // Step 8. The SS transmits UMD PDU#8 containing last segment of RLC SDU#4.

  // Anite : Changed from tsc_FI_StartOfSDU to tsc_FI_EndOfSDU as this is the last part of SDU 4

  f_TxUMD_PDUpartofSDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU4, 75, v_RLCSDUsize - 75, tsc_FI_EndOfSDU); // NOTE 1
  // Step 9. Wait for 200 ms to ensure that t_Reordering for the UMD PDU#2 expires.

  f_Delay (0.18);
  // Step 10. Check: Does the UE loop back the RLC SDU#2.

  // SDU tsc_RLC_SDU5 is the SDU 2 with PDCP SN = 0.
  f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU5 ); // NOTE3
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  // Step 11. Wait for 200 ms to ensure that t_Reordering for the UMD PDU#6 expires.

  f_Delay (0.18);  // NOTE 2
  // Step 12. Check: Does the UE loop back the RLC SDU#4.

  // SDU tsc_RLC_SDU6 is the SDU 4 with PDCP SN = 1.
  f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU6 ); // NOTE 3
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

} // fl_7_2_2_9_TestBody

	function fl_GenerateRLC_UM_SDUs_7229 (inout RLC_SS_State p_RLC_Rec, integer p_PDCP_SDU_size, integer p_SDU_Num)

{

  var integer i;

  for(i := 0; i < p_SDU_Num; i:= i + 1) {

    p_RLC_Rec.TxDataSize := p_PDCP_SDU_size - 1;

    f_GetTx_PRBS(p_RLC_Rec);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (

//        cs_PDCP_DataShortSN(int2bit(i, tsc_PDCP_SN_Size_7 ), p_RLC_Rec.RLC_Data));

//  temporary solution

        cs_PDCP_DataLongSN(i, p_RLC_Rec.RLC_Data));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_7;

    // SDU 2 will be the first PDCP PDU to be send hence the PDCP SEquence number should be zero.

    if(i == 1) {

    p_RLC_Rec.RLC_DataList[4] := f_PDCP_PDU_Encvalue (

        cs_PDCP_DataLongSN(i-1, p_RLC_Rec.RLC_Data));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_7;

    }

    // SDU 4 will be the second PDCP PDU to be send hence the PDCP SEquence number should be one. 

    if(i == 3) {

    p_RLC_Rec.RLC_DataList[5] := f_PDCP_PDU_Encvalue (

        cs_PDCP_DataLongSN(i-2, p_RLC_Rec.RLC_Data));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_7;

    } 

  }

}


5 Execution Log Files

5.1 LG Electronics LEO3 UE

The LG electronics LEO3 UE passed this test case on the Anite LTE solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_2_2_9_LG-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s100015: This archive comprises text format execution log file and the TTCN file.

